Newsgroup: it.comp.lang.visual-basic
Titolo: Inet e connessione internet

Ciao a tutti
Di solito vedo chiedere come fare a far partire la connessione internet se
questa non è attiva.
Io ho il problema inverso :)
Sto facendo un programmino simile al windows update, che ogni tanto va a
vedere se ci sono aggiornamenti disponibili.
Il problema è che se non sono connesso ad internet e con inet vado a vedere
se c'è qualcosa di nuovo mi parte la finestrella di connessione.
Come potrei fare ad evitare questo fastidioso comportamento?
Ciao, grazie
-- 
NeoDataType : 
http://www.neodatatype.net
Sito Comune : 
http://www.it-lang-vb.net

Ciao, grazie
La prima cosa che mi viene in mente: Non puoi fare prima un ping?
Avevo trovato una dll che lo faceva (ping.dll se non sbaglio).

 
Puoi provare a usare questa API:
http://www.mentalis.org/apilist/IsNetworkAlive.shtml
ciao
	maxx
-- 
NOSPAM: Rimuovere i trattini dallo username!
Cut hyphens from my username!

 Come potrei fare ad evitare questo fastidioso comportamento?
Fai un ping. Io uso questo modulo in Sinapsi Antispam (e poi gli passi il
Ping come indirizzo "
nome.host.chevuoi.it
"):
Option Explicit
''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
' Copyright ©1996-2004 VBnet, Randy Birch, All Rights Reserved.
' Some pages may also contain other copyrights by the author.
''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
' Distribution: You can freely use this code in your own
'               applications, but you may not reproduce
'               or publish this code on any web site,
'               online service, or distribute as source
'               on any media without express permission.
''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''
Private Const IP_STATUS_BASE As Long = 11000
Private Const IP_SUCCESS As Long = 0
Private Const IP_BUF_TOO_SMALL As Long = (11000 + 1)
Private Const IP_DEST_NET_UNREACHABLE As Long = (11000 + 2)
Private Const IP_DEST_HOST_UNREACHABLE As Long = (11000 + 3)
Private Const IP_DEST_PROT_UNREACHABLE As Long = (11000 + 4)
Private Const IP_DEST_PORT_UNREACHABLE As Long = (11000 + 5)
Private Const IP_NO_RESOURCES As Long = (11000 + 6)
Private Const IP_BAD_OPTION As Long = (11000 + 7)
Private Const IP_HW_ERROR As Long = (11000 + 8)
Private Const IP_PACKET_TOO_BIG As Long = (11000 + 9)
Private Const IP_REQ_TIMED_OUT As Long = (11000 + 10)
Private Const IP_BAD_REQ As Long = (11000 + 11)
Private Const IP_BAD_ROUTE As Long = (11000 + 12)
Private Const IP_TTL_EXPIRED_TRANSIT As Long = (11000 + 13)
Private Const IP_TTL_EXPIRED_REASSEM As Long = (11000 + 14)
Private Const IP_PARAM_PROBLEM As Long = (11000 + 15)
Private Const IP_SOURCE_QUENCH As Long = (11000 + 16)
Private Const IP_OPTION_TOO_BIG As Long = (11000 + 17)
Private Const IP_BAD_DESTINATION As Long = (11000 + 18)
Private Const IP_ADDR_DELETED As Long = (11000 + 19)
Private Const IP_SPEC_MTU_CHANGE As Long = (11000 + 20)
Private Const IP_MTU_CHANGE As Long = (11000 + 21)
Private Const IP_UNLOAD As Long = (11000 + 22)
Private Const IP_ADDR_ADDED As Long = (11000 + 23)
Private Const IP_GENERAL_FAILURE As Long = (11000 + 50)
Private Const MAX_IP_STATUS As Long = (11000 + 50)
Private Const IP_PENDING As Long = (11000 + 255)
Private Const PING_TIMEOUT As Long = 3000
Private Const WS_VERSION_REQD As Long = &H101
Private Const MIN_SOCKETS_REQD As Long = 1
Private Const SOCKET_ERROR As Long = -1
Private Const INADDR_NONE As Long = &HFFFFFFFF
Private Const MAX_WSADescription As Long = 256
Private Const MAX_WSASYSStatus As Long = 128
Private Type WSADATA
    wVersion As Integer
    wHighVersion As Integer
    szDescription(0 To MAX_WSADescription) As Byte
    szSystemStatus(0 To MAX_WSASYSStatus) As Byte
    wMaxSockets As Long
    wMaxUDPDG As Long
    dwVendorInfo As Long
End Type
Private Type ICMP_OPTIONS
    Ttl             As Byte
    Tos             As Byte
    Flags           As Byte
    OptionsSize     As Byte
    OptionsData     As Long
End Type
Public Type ICMP_ECHO_REPLY
    Address         As Long
    status          As Long
    RoundTripTime   As Long
    DataSize        As Long
    'Reserved        As Integer
    DataPointer     As Long
    Options         As ICMP_OPTIONS
    Data            As String * 250
End Type
Private Type HOSTENT
    hName As Long
    hAliases As Long
    hAddrType As Integer
    hLen As Integer
    hAddrList As Long
End Type
Private Declare Function gethostbyname Lib "wsock32" _
        (ByVal hostname As String) As Long
Private Declare Sub CopyMemory Lib "kernel32" _
        Alias "RtlMoveMemory" _
        (xDest As Any, _
        xSource As Any, _
        ByVal nbytes As Long)
Private Declare Function lstrlenA Lib "kernel32" _
        (lpString As Any) As Long
Private Declare Function WSAStartup Lib "wsock32" _
        (ByVal wVersionRequired As Long, _
        lpWSADATA As WSADATA) As Long
Private Declare Function WSACleanup Lib "wsock32" () As Long
Private Declare Function IcmpCreateFile Lib "icmp.dll" () As Long
Private Declare Function IcmpCloseHandle Lib "icmp.dll" _
        (ByVal IcmpHandle As Long) As Long
Private Declare Function IcmpSendEcho Lib "icmp.dll" _
        (ByVal IcmpHandle As Long, _
        ByVal DestinationAddress As Long, _
        ByVal RequestData As String, _
        ByVal RequestSize As Long, _
        ByVal RequestOptions As Long, _
        ReplyBuffer As ICMP_ECHO_REPLY, _
        ByVal ReplySize As Long, _
        ByVal Timeout As Long) As Long
Private Declare Function inet_addr Lib "wsock32" _
        (ByVal s As String) As Long
Public Function Ping(sAddress As String, _
                     sDataToSend As String, _
                     ECHO As ICMP_ECHO_REPLY) As Long
'If Ping succeeds :
'.RoundTripTime = time in ms for the ping to complete,
'.Data is the data returned (NULL terminated)
'.Address is the Ip address that actually replied
'.DataSize is the size of the string in .Data
'.Status will be 0
'
'If Ping fails .Status will be the error code
On Error GoTo errore:
    Dim hPort As Long
    Dim dwAddress As Long
    'convert the address into a long representation
    dwAddress = inet_addr(sAddress)
    'if dwAddress is valid
    If dwAddress <> INADDR_NONE Then
        'open a port
        hPort = IcmpCreateFile()
        'and if successful,
        If hPort Then
            'ping it.
            Call IcmpSendEcho(hPort, _
                    dwAddress, _
                    sDataToSend, _
                    Len(sDataToSend), _
                    0, _
                    ECHO, _
                    Len(ECHO), _
                    PING_TIMEOUT)
            'return the status as ping success
            Ping = ECHO.status
            'close the port handle
            Call IcmpCloseHandle(hPort)
        End If    'If hPort
    Else:
        'the address format was probably invalid
        Ping = INADDR_NONE
    End If
Exit Function
errore:
    FrmMain.Text2.Text = FrmMain.Text2.Text & Err.number & ": " &
Err.Description & vbCrLf
    FrmMain.Text2.SelStart = Len(FrmMain.Text2.Text)
End Function
Public Function GetStatusCode(status As Long) As String
On Error GoTo errore:
    Dim MSG As String
    Select Case status
        Case IP_SUCCESS: MSG = "ip success"
        Case INADDR_NONE: MSG = "inet_addr: bad IP format"
        Case IP_BUF_TOO_SMALL: MSG = "ip buf too_small"
        Case IP_DEST_NET_UNREACHABLE: MSG = "ip dest net unreachable"
        Case IP_DEST_HOST_UNREACHABLE: MSG = "ip dest host unreachable"
        Case IP_DEST_PROT_UNREACHABLE: MSG = "ip dest prot unreachable"
        Case IP_DEST_PORT_UNREACHABLE: MSG = "ip dest port unreachable"
        Case IP_NO_RESOURCES: MSG = "ip no resources"
        Case IP_BAD_OPTION: MSG = "ip bad option"
        Case IP_HW_ERROR: MSG = "ip hw_error"
        Case IP_PACKET_TOO_BIG: MSG = "ip packet too_big"
        Case IP_REQ_TIMED_OUT: MSG = "ip req timed out"
        Case IP_BAD_REQ: MSG = "ip bad req"
        Case IP_BAD_ROUTE: MSG = "ip bad route"
        Case IP_TTL_EXPIRED_TRANSIT: MSG = "ip ttl expired transit"
        Case IP_TTL_EXPIRED_REASSEM: MSG = "ip ttl expired reassem"
        Case IP_PARAM_PROBLEM: MSG = "ip param_problem"
        Case IP_SOURCE_QUENCH: MSG = "ip source quench"
        Case IP_OPTION_TOO_BIG: MSG = "ip option too_big"
        Case IP_BAD_DESTINATION: MSG = "ip bad destination"
        Case IP_ADDR_DELETED: MSG = "ip addr deleted"
        Case IP_SPEC_MTU_CHANGE: MSG = "ip spec mtu change"
        Case IP_MTU_CHANGE: MSG = "ip mtu_change"
        Case IP_UNLOAD: MSG = "ip unload"
        Case IP_ADDR_ADDED: MSG = "ip addr added"
        Case IP_GENERAL_FAILURE: MSG = "ip general failure"
        Case IP_PENDING: MSG = "ip pending"
        Case PING_TIMEOUT: MSG = "ping timeout"
        Case Else: MSG = "unknown  msg returned"
    End Select
    GetStatusCode = CStr(status) & "   [ " & MSG & " ]"
Exit Function
errore:
    FrmMain.Text2.Text = FrmMain.Text2.Text & Err.number & ": " &
Err.Description & vbCrLf
    FrmMain.Text2.SelStart = Len(FrmMain.Text2.Text)
End Function
Public Function GetIPFromHostName(ByVal sHostName As String) As String
On Error GoTo errore
'converts a host name to an IP address.
    Dim nbytes As Long
    Dim ptrHosent As Long    'address of hostent structure
    Dim ptrName As Long    'address of name pointer
    Dim ptrAddress As Long    'address of address pointer
    Dim ptrIPAddress As Long
    Dim sAddress As String
    sAddress = Space$(4)
    ptrHosent = gethostbyname(sHostName & vbNullChar)
    If ptrHosent <> 0 Then
        'assign pointer addresses and offset
        'ptrName is the official name of the host (PC).
        'If using the DNS or similar resolution system,
        'it is the Fully Qualified Domain Name (FQDN)
        'that caused the server to return a reply.
        'If using a local hosts file, it is the first
        'entry after the IP address.
        ptrName = ptrHosent
        'Null-terminated list of addresses for the host.
        'The Address is offset 12 bytes from the start of
        'the HOSENT structure. Note: Here we are retrieving
        'only the first address returned. To return more than
        'one, define sAddress as a string array and loop through
        'the 4-byte ptrIPAddress members returned. The last
        'item is a terminating null. All addresses are returned
        'in network byte order.
        ptrAddress = ptrHosent + 12
        'get the IP address
        CopyMemory ptrName, ByVal ptrName, 4
        CopyMemory ptrAddress, ByVal ptrAddress, 4
        CopyMemory ptrIPAddress, ByVal ptrAddress, 4
        CopyMemory ByVal sAddress, ByVal ptrIPAddress, 4
        GetIPFromHostName = IPToText(sAddress)
    End If
Exit Function
errore:
    FrmMain.Text2.Text = FrmMain.Text2.Text & Err.number & ": " &
Err.Description & vbCrLf
    FrmMain.Text2.SelStart = Len(FrmMain.Text2.Text)
End Function
Private Function IPToText(ByVal IPAddress As String) As String
    IPToText = CStr(Asc(IPAddress)) & "." & _
            CStr(Asc(Mid$(IPAddress, 2, 1))) & "." & _
            CStr(Asc(Mid$(IPAddress, 3, 1))) & "." & _
            CStr(Asc(Mid$(IPAddress, 4, 1)))
End Function
Public Sub SocketsCleanup()
    If WSACleanup() <> 0 Then
        FrmMain.Text2.Text = FrmMain.Text2.Text & "Windows Sockets error
occurred in Cleanup." & vbCrLf
        FrmMain.Text2.SelStart = Len(FrmMain.Text2.Text)
    End If
End Sub
Public Function SocketsInitialize() As Boolean
    Dim WSAD As WSADATA
    SocketsInitialize = WSAStartup(WS_VERSION_REQD, WSAD) = IP_SUCCESS
End Function
'--end block--
-- 
"Seminate Linux, che Windows si pianta da sè...." (Antico Adagio)
Gestione Ottica - Gestionale per Ottici e Programmi Gratuiti su:
http://www.nicolaottomano.it
Sinapsi Antispam. Antispam Gratuito.
http://www.nicolaottomano.it/public/sinapsi.htm
email spamtrap: 
spam...@nicolaottomano.it
Velocipedi Equestri su
http://www.nicolaottomano.it/velocipedi_equestri.htm